>

MATH 392 -- Seminar in Computational Commutative Algebra
Singular Points of Varieties
March 13, 2019

A large part of modern algebraic geometry deals with the study of *singular points of varieties* or singularities for short. For example, Let's consider again the envelope of the family of circles from question II on this week's lab/problem set:

> $F := (x - t)^2 + (y - t^2)^2 - 1;$

$$F := (x-t)^2 + (-t^2+y)^2 - 1 \qquad \textbf{(1)}$$

> $Id := [F, diff(F, t)];$

$$Id := \left[(x-t)^2 + (-t^2+y)^2 - 1,\ -2x+2t-4(-t^2+y)\,t\right] \qquad \textbf{(2)}$$

> $with(Groebner):$

> $BEnv := Basis(Id, plex(t, x, y));$

$$\begin{aligned} BEnv := [&16x^6+16x^4y^2-40x^4y-32x^2y^3-47x^4+16y^4+6x^2y-40y^3 \\ &+28x^2+9y^2+40y-25,\ 128x^5y+128x^3y^3+256ty^4+224x^5-224x^3y^2 \\ &-384xy^4-128ty^3-712x^3y-160xy^3-192ty^2-738x^3-24xy^2 \\ &-488ty+354yx+535t+197x,\ 64txy^3-32x^4y-32x^2y^3+48txy^2 \\ &+16x^4+32x^2y^2-32y^4+12txy+22x^2y-16y^3-107tx+39x^2+102y^2 \\ &+16y-70,\ -32x^5-32x^3y^2-64ty^3+112x^3y+96xy^3+144tx^2 \\ &+144ty^2-18x^3-128xy^2-60ty-58yx-25t+85x,\ 32txy^2-16x^4 \\ &-16x^2y^2+40txy+8x^2y-16y^3+27t^2+26tx+15x^2-16y^2+16y \\ &+16] \end{aligned} \qquad \textbf{(3)}$$

The envelope is the variety defined by the first polynomial here -- the generator of the first elimination ideal. Plotting with a 200×200 grid of sample points, we see that there is some fine structure around the crossings with the y-axis. In particular, there are perhaps (at least) three points where the envelope curve appears to lack a single *well-defined tangent line:*

> $with(plots):$

> $implicitplot(BEnv[1], x = -1..1, y = -1..5, grid = [200, 200], scaling = constrained);$



Recall from Multivariable Calculus (MATH 241), that for an implicit curve in the plane given by an equation $f(x, y) = 0$, the tangent line at P = (x_0, y_0) is defined by the equation

$$\frac{\partial}{\partial x} f(x\_0, y\_0) \cdot (x - x0) + \frac{\partial}{\partial y} f(x\_0, y\_0) \cdot (y - y\_0) = 0$$

and this gives a well-defined line in the plane passing through P as long as at least one of the partial derivatives of *f* is nonzero at P.

Hence, we are led to the following (provisional) definition:

Definition. *Let* $C = V(f)$ *be a variety in* k^2. *A point* $P = (x\_0, y\_0) \in C$ *is said to be a* ***singular point*** *of* C *if* $\frac{\partial}{\partial x} f(x\_0, y\_0) = \frac{\partial}{\partial y} f(x\_0, y\_0) = 0.$

For a first example consider $f(x, y) = x^2 + \frac{y^2}{4} - 1$

> $fEll := x^2 + \frac{y^2}{4} - 1;$

$$fEll := x^2 + \frac{y^2}{4} - 1 \quad (4)$$

> $Basis([fEll, diff(fEll, x), diff(fEll, y)], plex(x, y));$

$$[1] \quad (5)$$

>

Question: What does this result mean algebraically? Geometrically?

For a second example, consider

$V(9x^4+18x^2y^2+9y^4-18x^2y+6y^3-152x^2-152y^2+512)$,

plotted below:

> $implicitplot(9x^4+18x^2y^2+9y^4-18x^2y+6y^3-152x^2-152y^2+512, x=-4..4, y=-4..4, grid=[100, 100]);$



The singular points would be the points where the curve appears to cut through itself. To find their coordinates, we solve the system of three equations consisting of the equation $f(x, y)$ and its two partial derivatives:

> $f := 9x^4+18x^2y^2+9y^4-18x^2y+6y^3-152x^2-152y^2+512;$

$$f := 9x^4+18x^2y^2+9y^4-18x^2y+6y^3-152x^2-152y^2+512 \quad (6)$$

> $Sings := [f, diff(f, x), diff(f, y)];$

$$Sings := [9x^4+18x^2y^2+9y^4-18x^2y+6y^3-152x^2-152y^2+512, 36x^3+36xy^2-36xy-304x, 36x^2y+36y^3-18x^2+18y^2-304y] \quad (7)$$

> $BSings := Basis(Sings, plex(x, y));$

$$BSings := [3x^3-16x, 9x^2+12y-32] \quad (8)$$

> $factor(BSings[1]);$

$$(3y+4)(3y-8) \quad (9)$$

> $subs\left(y=-\frac{4}{3}, BSings\right);$

$$[0, 0, 9x^2 - 48] \quad \textbf{(10)}$$

> $subs\left(y = \frac{8}{3}, BSings\right);$

$$[0, 12x, 9x^2] \quad \textbf{(11)}$$

These computations show that the singular points are:

$$(x, y) = \left(\pm 4\frac{\sqrt{3}}{3}, -\frac{4}{3}\right), \left(0, \frac{8}{3}\right)$$

These singularities would all be called *simple nodes* (they're just like origin on the nodal cubic $V(y^2 - x^2 - x^3)$), and the curve C would be called a *trinodal quartic* because there are three such points on C (all visible in the real picture!). One interesting question to ask is: How many, and what kinds of, singular points can a curve defined by a polynomial of a given total degree have? A lot of ***beautiful*** stuff is known there, but unfortunately we won't be able to pursue it in our class except in a few examples :(

Let's return to our original envelope curve for the family of circles. We can ask: how many and what kinds of singular points does it have?

> $EnvSings := [BEnv[1], diff(BEnv[1], x), diff(BEnv[1], y)];$

$$EnvSings := [16x^6 + 16x^4y^2 - 40x^4y - 32x^2y^3 - 47x^4 + 16y^4 + 6x^2y - 40y^3 + 28x^2 + 9y^2 + 40y - 25, 96x^5 + 64x^3y^2 - 160x^3y - 64xy^3 - 188x^3 + 12yx + 56x, 32x^4y - 40x^4 - 96x^2y^2 + 64y^3 + 6x^2 - 120y^2 + 18y + 40] \quad \textbf{(12)}$$

> $BEnvSings := Basis(EnvSings, plex(x, y));$

$$BEnvSings := [16384y^7 + 4096y^6 - 15360y^5 - 69376y^4 + 22208y^3 + 40368y^2 + 58636y - 57245, 4096y^6x + 6144xy^5 + 3840xy^4 - 12544xy^3 - 10128xy^2 - 2568yx + 11449x, 4096y^6 - 6144y^5 - 11520y^4 - 20224y^3 + 46656x^2 + 70896y^2 - 7992y - 30035] \quad \textbf{(13)}$$

> $factor(BEnvSings[1]);$

$$(-5 + 4y)(64y^3 + 48y^2 + 12y - 107)^2 \quad \textbf{(14)}$$

> $subs\left(y = \frac{5}{4}, BEnvSings\right);$

$$[0, 11664x, 46656x^2] \quad \textbf{(15)}$$

Note that this says there is one singular point with $y = \frac{5}{4}$ and $x = 0$. As you can see if you look back at the plot of the variety defined by BEnv[1] = 0, this is also a simple node.

Interestingly, all the other roots of the equation BEnvSings[1] = 0 are double roots(!) The cubic in y appearing in the factorization above has one real root and one complex conjugate pair:

> $Digits := 20:$

> $fsolve(64\,y^3 + 48\,y^2 + 12\,y - 107, y, complex);$

$$-0.84527539448807480303 - 1.0310472277489519698\,I,\ -0.84527539448807480303 + 1.0310472277489519698\,I,\ 0.94055078897614960606 \quad \textbf{(16)}$$

> $subs(y = 0.94055078897614960606, BEnvSings);$

$$[1.\,10^{-15},\ 1.\,10^{-15}\,x,\ 46656\,x^2 - 2364.022854628432458] \quad \textbf{(17)}$$

> $fsolve(46656\,x^2 - 2364.022854628432458, x);$

$$-0.22509823218728275682,\ 0.22509823218728275682 \quad \textbf{(18)}$$

These two points
$(x, y) = (\pm 0.22509823218728275682, 0.94055078897614960606)$
(approximately) are the other two real singular points visible in the plot from before.

To finish, let's see that these are of a different type than the nodes seen in the other examples. To do this, let's translate our curve so that the origin is located at the singular
point. First for the point at $\left(0, \frac{5}{4}\right)$:

> $Trans1 := expand\left(subs\left(x = X, y = Y + \frac{5}{4}, BEnv[1]\right)\right);$

$$Trans1 := 16\,X^6 + 16\,X^4\,Y^2 - 32\,X^2\,Y^3 - 72\,X^4 - 120\,X^2\,Y^2 + 16\,Y^4 - 144\,X^2\,Y + 40\,Y^3 - 27\,X^2 + 9\,Y^2 \quad \textbf{(19)}$$

Note that here, the terms of lowest total degree are:

$$9\,Y^2 - 27 \cdot X^2 = \left(3 \cdot Y - 3\sqrt{3} \cdot X\right) \cdot \left(3\,Y + 3 \cdot \sqrt{3}\,X\right).$$

The two distinct linear factors tell us that we are dealing with a simple node -- they define the tangent lines to the two "branches" of the curve passing through the node (in the translated coordinate system).

Something different happens at each of the other points but we actually have very careful in the computation to see what is happening:

> $Digits := 20;$

$$Digits := 20 \quad \textbf{(20)}$$

> $Trans2 := expand(subs(x = X + 0.22509823218728275682, y = Y + 0.94055078897614960606, BEnv[1]));$

$$Trans2 := -59.798985754242026219\,X^3 + 16\,X^4\,Y^2 - 32\,X^2\,Y^3 - 85.428631184861735725\,X^2\,Y^2 - 81.935508798478853206\,X^2\,Y + 16\,X^6 - 58.307247580625897983\,X^4 - 13.789269245272822876\,X^2 \quad \textbf{(21)}$$

$$- 23.475050306888694176\, Y^2 + 21.609430289979144654\, X^5$$
$$+ 18.573835642190699303\, Y^3 + 16\, Y^4 - 14.406286859986096436\, X Y^3$$
$$- 8.916028205211921261\, X^3 Y - 39.919578183347820789\, X Y^2$$
$$- 35.983540083099806810\, X Y - 9.902374752763212606\, X^4 Y$$
$$+ 14.406286859986096436\, X^3 Y^2 - 2.\, 10^{-18}\, Y$$

We need to treat the Y term as zero here (they would be exactly zero if we had the exact coordinates of these points rather than the decimal approximations). The nonzero terms of lowest total degree are:

> $Q := -13.789269245272822876\, X^2 - 23.475050306888694176\, Y^2$
> $\quad -35.983540083099806810\ X Y;$

$$Q := -13.789269245272822876\, X^2 - 23.475050306888694176\, Y^2 - 35.983540083099806810\, X Y \tag{22}$$

> $fsolve(subs(X = 1, Q), Y, complex);$

$$-0.76642093654087991296 - 3.8562580974251994358\, 10^{-10}\, \mathrm{I}, \quad -0.76642093654087991296 + 3.8562580974251994358\, 10^{-10}\, \mathrm{I} \tag{23}$$

The approximate imaginary parts of these roots are extremely small -- the exact value is actually 0. That means that this Q is actually a perfect square. These two singular points are ***simple cusps.***

>